# Simultaneous eqns in maple

I am trying to solve the following system of equations in maple but it doesn't work for some reason:

````solve({
a*(1-x)-x*f-x*e = 0, 
b*(1-x)-x*c-x*d = 0, 
c*(1-z)-z*b-z*a = 0, 
d*(1-z)-z*e-z*f = 0, 
e*(1-y)-y*d-y*c = 0, 
f*(1-y)-y*a-y*b = 0, 
a+b+c+d+e+f-1 = 0 },
{a, b, c, d, e, f
})
````
-

I tried just now and for me, no answers in outputs, so, no solutions. – Sigur Sep 4 '12 at 2:07

## 2 Answers

This linear system of equations is inconsistent. One way to see this is to recognize that the first 6 equations imply that the variables `a` to `f` all have value zero. But the last equation dictates that their sum is equal to 1. Clearly, if they are all zero then they cannot add up to 1.

````eqs:=[a*(1-x)-x*f-x*e = 0, 
  b*(1-x)-x*c-x*d = 0, 
  c*(1-z)-z*b-z*a = 0, 
  d*(1-z)-z*e-z*f = 0, 
  e*(1-y)-y*d-y*c = 0, 
  f*(1-y)-y*a-y*b = 0, 
  a+b+c+d+e+f-1 = 0]:

vars:=[a, b, c, d, e, f]:

with(LinearAlgebra):
````

Now compare results from

````linsys:=GenerateMatrix(eqs[1..6],vars,augmented);
LinearSolve(linsys);

LUDecomposition(GenerateMatrix(eqs[1..6],vars,augmented),output=R);
%[1..-1,1..6].Vector(vars)=%[1..-1,7];
````

with that from,

````linsys:=GenerateMatrix(eqs,vars,augmented);
LinearSolve(linsys);

LUDecomposition(GenerateMatrix(eqs,vars,augmented),output=R);
%[1..-1,1..6].Vector(vars)=%[1..-1,7];
````
-

The equations look linear in $a,b,c,d,e,f$. As such, you can use $$A,b:=LinearAlgebra[GenerateMatrix](system\_of\_equations,variables);$$ and then $LinearAlgebra[LinearSolve](A,b)$ to solve the matrix system.

-
{"extraction_info": {"found_math": true, "script_math_tex": 0, "script_math_asciimath": 0, "math_annotations": 0, "math_alttext": 0, "mathml": 0, "mathjax_tag": 0, "mathjax_inline_tex": 2, "mathjax_display_tex": 1, "mathjax_asciimath": 0, "img_math": 0, "codecogs_latex": 0, "wp_latex": 0, "mimetex.cgi": 0, "/images/math/codecogs": 0, "mathtex.cgi": 0}, "config": {"markdown_headings": true, "markdown_code": true, "boilerplate_config": {"ratio_threshold": 0.18, "absolute_threshold": 10, "end_threshold": 15, "enable": true}, "remove_buttons": true, "remove_image_figures": true, "remove_link_clusters": true, "table_config": {"min_rows": 2, "min_cols": 3, "format": "github"}, "remove_chinese": true, "remove_edit_buttons": true, "extract_latex": true}, "english_confidence": 0.7296240925788879, "perplexity_flag": "middle"}